
Работа с файлами.
Под файлом понимается именованная область внешней памяти ПК (жёсткого диска, гибкой дискеты, электронного «виртуального» диска)

Любой файл имеет три характерные особенности. Во-первых, у него есть имя, что дает возможность программе работать одновременно с несколькими файлами. Во-вторых, он содержит компоненты одного типа. Типом компонентов может быть любой тип Турбо Паскаля, кроме файлов. Иными словами, нельзя создать «файл файлов». В—третьих, длина вновь создаваемого файла никак не оговаривается при его объявлении и ограничивается только емкостью устройств внешней памяти. 

Файловый тип или переменную файлового типа можно задать одним из трёх способов:

<имя> = FILE OF <тип>;

<имя> = TEXT;

<имя> = FILE;

Здесь <имя> - имя файлового типа (правильный идентификатор);

FILE, OF — зарезервированные слова (файл, из); 

TEXT — имя стандартного типа текстовых файлов;

<тип> — любой тип Турбо Паскаля, кроме файлов.

Например:

type 

 product = record 

  name : string;

  code : word;

  cost : comp 

 end;

 text 80 = file of string [80];

var

 f1 : file of char; 

 f2 : text; 

 f3 : file;

 f4 : text 80; 

 f5 : file of product;

В зависимости от способа объявления можно выделить три вида файлов: 

· типизированные файлы (задаются предложением FILE OF...);
· текстовые файлы (определяются типом TEXT);
· нетипизированные файлы (определяются типом FILE).

(F1, F4 и F5— типизированные файлы, F2 — текстовый файл, F3 — нетипизированный файл)

ДОСТУП К ФАЙЛАМ

Файлы становятся доступны программе только после выполнения особой процедуры открытия файла. Эта процедура заключается в связывании ранее объявленной файловой переменной с именем существующего или вновь создаваемого файла, а также в указании направления обмена ин​формацией: чтение из файла или запись в него.

Файловая переменная связывается с именем файла в результате об​ращения к стандартной процедуре ASSIGN :

ASSIGN (<ф.п.>, <имя файла>);

Здесь <ф.п.> — файловая переменная (правильный идентификатор, объявленный в программе как переменная файлового типа);

<имя файла или л.у.> — текстовое выражение, содержащее имя файла.

Вспомним:

Имя файла — это любое выражение строкового типа, которое строится по правилам определения имен в MS-DOS (операционной системе ПК):

· имя содержит до восьми разрешенных символов; разрешенные символы — это прописные и строчные латинские буквы, цифры и символы: ! @ # $ % ^ & ( ) ' - _

· за именем может следовать расширение — последовательность до трех разрешенных символов; расширение, если оно есть, отделяется от имени точкой.

Перед именем может указываться так называемый путь к файлу: имя диска и/или имя текущего каталога и имена каталогов вышестоящих уровней. Максимальная длина имени вместе с путем — 79 символов, например:

var 

  finp : text;

  fout : file of string;

const

  name = 'c:\dir\subdir\out.txt';

begin

  assign (finp, '123.dat' ) ;

  assign (fout, name); 

end.

ИНИЦИАЦИЯ ФАЙЛА
Инициировать файл означает указать для этого файла направление передачи данных. В Турбо Паскале можно открыть файл для чтения, для записи информации, а также для чтения и записи одновременно.

Для чтения файл инициируется с помощью стандартной процедуры RESET:                   RESET (<ф.п.>);

Здесь <ф.п.> — файловая переменная, связанная ранее процедурой ASSIGN с уже существующим файлом или логическим устройством — приемником информации.

Для текстовых файлов, открытых процедурой RESET, нельзя использовать процедуру WRITE или WRITELN.

Стандартная процедура

REWRITE (<ф.п.>);

инициирует запись информации в файл или в логическое устройств связанное ранее с файловой переменной <ф.п.>. Процедурой REWRITE нельзя инициировать запись информации в ранее существовавший дисковый файл: при выполнении этой процедуры старый файл уничтожает и никаких сообщений об этом в программу не передается. Новый файл подготавливается к приему информации и его указатель принимает значение 0.

Стандартная процедура

APPEND (<ф.п.>);

инициирует запись в ранее существовавший текстовый файл для его расширения, при этом указатель файла устанавливается  в его конец. Процедура APPEND применима только к текстовым файлам. 

ПРОЦЕДУРЫ И ФУНКЦИИ ДЛЯ РАБОТЫ С ФАЙЛАМИ

Процедура Close. Закрывает файл, однако связь файловой переменной с именем файла, установленная ранее процедурой ASSIGN, сохраняется.

Процедура Rename. Переименовывает файл.
Процедура Erase.  Уничтожает файл.
Процедура FLUSH. Очищает внутренний буфер файла и, таким образом, гарантирует сохранность всех последних изменений файла на диске.

Функция EOF. Логическая функция, тестирующая конец файла. Возвращает TRUE, если файловый указатель стоит в конце файла. 

Процедура CHDIR. Изменение текущего каталога. 

Процедура GETDIR. Позволяет определить имя текущего каталога 

Процедура MKDIR. создает новый каталог на указанном диске.

Процедура RMDIR. удаляет каталог.

Функция IORESULT; WORD. Возвращает условный признак последней операции ввода—вывода. Если операция завершилась успешно, функция возвращает ноль. 

Функция DISKFREE «диск»; LONGINT. Возвращает объем в байтах свободного пространства на указанном диске. 

Функция DISKSIZE «диск»: LONGINT. Возвращает полный объем диска в байтах или —1 , если указан номер несуществующего диска.

Процедура FINDFIRST. Возвращает атрибуты первого из файлов, за​регистрированных в указанном каталоге.

Пропедура findnext. Возвращает имя следующего файла в каталоге.

Процедура GETFTIME. Возвращает время создания или последнего новления файла.

Процедура SETFTIME. Устанавливает новую дату создания или обнов— кия файла. 

Процедура getfattr. Позволяет получить атрибуты файла.

Процедура SETFATTR. Позволяет установить атрибуты файла. 

функция FSEARCH: PATHSTR. Ищет файл в списке каталогов.

ТЕКСТОВЫЕ ФАЙЛЫ

Текстовые файлы связываются с файловыми переменными, принадлежащими стандартному типу TEXT. Текстовые файлы предназначены для Хранения текстовой информации. Именно в такого типа файлах хранятся, Например, исходные тексты программ. Компоненты (записи) текстового файла могут иметь переменную длину, что существенно влияет на ха​рактер работы с ними.

Текстовый файл трактуется в Турбо Паскале как совокупность строк временной длины. Доступ к каждой строке возможен лишь последовательно, начиная с первой. При создании текстового файла в конце каждой строки ставится специальный признак EOLN (End Of LiNe — конец строки), а в конце всего файла — признак EOF {End Of File — конец файла).

Для доступа к записям применяются процедуры READ, READLN, WRiTp WRITELN.

Пропедура READ. Обеспечивает ввод символов, строк и чисел.

Процедура READLN. Обеспечивает ввод символов, строк и чисел. Эта

процедура идентична процедуре READ за исключением того, что после считывания последней переменной оставшаяся часть строки до маркера EOLN пропускается, поэтому следующее обращение к READLN или READ начинается с первого символа новой строки.

Процедура WRITELN. Обеспечивает вывод информации в текстовый файл. 

Процедура WRITELN.  Эта процедура полностью идентична процедуре WRITE за исключением того, что можно опускать параметр <сп.вывода>: в этом случае в файл передается маркер EOLN, что при выводе на экран приведет к переводу курсора в начало следующей строки.

Логическая функция EQLN. Возвращает TRUE, если во входном текстовом файле достигнут маркер конца строки. 
ТИПИЗИРОВАННЫЕ ФАЙЛЫ

Длина любого компонента типизированного файла строго постоянна, что дает возможность организовать прямой доступ к каждому из них (т.е. доступ к компоненту по его порядковому номеру).

Перед первым обращением к процедурам ввода — вывода указатель файла стоит в его начале и указывает на первый компонент с номером 0.

Процедура READ. Обеспечивает чтение очередных компонентов типи​зированного файла. 

Процедура write. Используется для записи данных в типизированный файл. 

Процедура SEEK. Смещает указатель файла к требуемому компоненту. Формат обращения:

SEEK «ф.п.>,<N компонента»

Здесь <N компонента> — выражение типа LONGINT, указывающее номер компонента файла.

Функция FILESIZE. Возвращает значение типа LONGINT, которое со​держит количество компонентов файла.

Функция FILEPOS. Возвращает значение типа LONGINT, содержащее порядковый номер компонента файла, который будет обрабатываться  следующей операцией ввода—-вывода. 

НЕТИПИЗИРОВАННЫЕ ФАЙЛЫ

Нетипизированные файлы объявляются как файловые переменные типа FILE и отличаются тем,  что для них не указан тип компонентов.  Отсутствие типа делает эти файлы, с одной стороны, совместимыми с любыми другими файлами, а с другой — позволяет организовать высокоскоростной обмен данными между памятью и диском. 

Задание 7-1.  Ввести с клавиатуры текст и сохранить его в виде файла “Proba.txt”  

program zadanie7_1;


  var f:text;

      s:string;

Begin 

  writeln('Введите текст');

  readln(s);

  assign(f,'proba.txt');  {связываем файловую переменную f с именем файла}

  rewrite(f);             {открываем файл для записи, файл при этом создаётся} 

  writeln(f,s);           {выводим информацию в файл} 

  close(f);               {закрываем файл}  

End.

Задание 7-2.  Ввести с клавиатуры текст и дописать его в существующий файл “Proba.txt”  

program zadanie7_2;


  var f:text;

      s:string;

Begin 

  writeln('Введите текст');

  readln(s);

  assign(f,'proba.txt');  {связываем файловую переменную f с именем файла}

  append(f);              {открываем файл для добавления} 

  writeln(f,s);           {выводим информацию в файл} 

  close(f);               {закрываем файл}  

End.

Задание 7-3.  Ввести на экран содержимое файла “Proba.txt”  

program zadanie7_3;


  var p:text;

      t:string;

Begin 

  assign(p,'proba.txt');  {связываем файловую переменную f с именем файла}

  reset (p);              {открываем файл для чтения}

  while not eof(p) do     {пока не конец файла} 



 begin

       readln(p,t);       {считываем из файла}  

       writeln(t);        {выводим информацию на экран} 

     end;

  close(p);               {закрываем файл}  

End.

Задание 7-4.  Сохранить элементы двумерного массива в файле под именем “massiv.in”  

program zadanie7_4;

var n,i,j:integer;

        f:text;

        a: array[1..50,1..50] of integer;

Begin

 writeln('Введите число N');

 readln(n);

 writeln('Введите элементы массива');

 for i:=1 to n do


 for j:=1 to n do


   
readln(a[i,j]);

 assign(f,'massiv.in'); rewrite(f);

 writeln(f,n);

 for i:=1 to n do

 
begin

for j:=1 to n do

   write(f,a[i,j],' ');




 writeln(f);

  end;

 close(f);

End.

Задание 7-5.  Считать элементы двумерного массива из  файла под именем “massiv.in”, причем в первой строке файла записана размерность массива.

program zadanie7_5;

var n,i,j : integer;

        f
: text; 

        a : array[1..50,1..50] of integer;

Begin

  assign(f,'massiv.in');

  reset(f);

  readln(f,n);             {из первой строки считываем размер массива}   

  for i:=1 to n do


  for j:=1 to n do

       read(f,a[i,j]);     {считываем весь массив}

  writeln('Считанный из файла массив: ');

  for i:=1 to n do


  
begin  

       for j:=1 to n do

           write(a[i,j],' '); {выводим массив на экран}





  writeln;



  end;

close(f);

End.

Задание 7-6. Создать файл ‘cisla.nat’ в который записать более 10 чисел. Десять первых чисел считать из файла в массив. Вывести массив на экран.

program zadacha7_6;

  const n=10;

  var   k,i,a:integer;

            f:text;

            m:array[1..n] of integer;

Begin

  assign(f,'cisla.nat');

  rewrite(f);                             {открываем файл для ЗАПИСИ}

  writeln('Введите кол-во чисел > 10');

  readln (k);

  for i:=1 to k do

     begin

        writeln('Введите ',i,' число');

        readln(a);

        writeln(f,a);                      {сохраняем в файл}

     end;

 close(f);

 reset (f);                                {открываем файл для ЧТЕНИЯ}

 i:=1;

 while not eof(f) and (i<=10) do           {пока не конец файла и i<10}

      begin

         readln(f,a);                      {считываем из файла}

         m[i]:=a;                          {сохраняем  в массив}

         inc(i);

     end;

  close(f);                                {закрываем файл}

  for i:=1 to 10 do

        write(m[i],' ');

End.

Задание 7-6. Из файла 'cisla.nat' перенести в файл 'chet.cis' все чётные числа. Вывести содержимое файла  'chet.cis' на экран. 

program zadacha7_7;

var     a:integer;

    f1,f2:text;

Begin

  assign(f1,'cisla.nat');

  assign(f2,'chet.cis');

  reset (f1);                             {открываем файл для чтения}

  rewrite(f2);                            {открываем файл для записи}

  while not eof(f1) do                    {пока не конец файла}

   begin

    readln(f1,a);                        {считываем из файла f1}

    if a mod 2 =0 then writeln(f2,a); {если чётное, то сохраняем в файл f2}

   end;

 close(f1);close(f2);                   {закрываем файл}

               {выводим содержимое файла 'chet.cis'}

  reset (f2);                           {открываем файл для чтения}

  while not eof(f2) do

    begin

     readln(f2,a);                      {считываем из файла f2}

     writeln(a);

    end;

  close(f2);

End.

